





# Formal verification of an algorithm for computing strongly connected components

Vincent Trélat Supervized by Stephan Merz

June 1, 2022

My warmest and most sincere thanks to Stephan Merz, who has been a great help in the development the whole project and who reviewed this paper with great care.

## Contents

1	$\mathbf{Pre}$	eamble 3					
	1.1	Academic context					
	1.2	Formal methods					
	1.3	Isabelle (HOL)					
2	Inti	roduction					
3	For	malisation					
	3.1	Graphs and reachability					
	3.2	Strongly connected components					
		3.2.1 Directed graphs					
		3.2.2 Examples					
4	$\mathbf{A} \mathbf{s}$	A sequential set-based algorithm 6					
	4.1	Formalisation					
	4.2	The algorithm					
	4.3	Informal proof					
	4.4	Prerequisites for the formal proof					
		4.4.1 Environment setup					
		4.4.2 Reachability					
		4.4.3 Ordering relation					
		4.4.4 Implementation of the algorithm					
		4.4.5 General scheme					
		4.4.6 Well-formedness of the environment					
		4.4.7 dfs pre- and post-conditions					
		4.4.8 dfss pre- and post-conditions					
	4.5	Formal proof					
		4.5.1 pre_dfs implies pre_dfss					
		4.5.2 pre_dfss implies pre_dfs					
		4.5.3 pre_dfs implies post_dfs					
		4.5.4 Partial correctness					
5	Cor	nclusion					
	5.1	What has been done?					
	5.2	What is missing?					
	5.3	My experience					
6	Apı	pendix 24					
	6.1	Some lemmas					
	6.2	An example of execution of algorithm 1					

## 1 Preamble

#### 1.1 Academic context

This research work was carried out as part of my curriculum at the French École des Mines de Nancy. All documents such as codes or source papers are available on a GitHub repository.

#### 1.2 Formal methods

Formal methods are a field of computer science related to mathematical logic and reasoning. The whole purpose of the discipline is to give precise, mathematical definitions to computer science concepts. Formal methods find applications in a variety of fields, both concrete, such as the railway industry or self-driving cars, and abstract, such as computational architecture. Although the purpose of giving such definitions is to enable formal verification, many techniques besides theorem proving, such as model-based testing, run-time monitoring, model checking etc. are used.

## 1.3 Isabelle (HOL)

"Isabelle is a generic proof assistant. It allows mathematical formulas to be expressed in a formal language and provides tools for proving those formulas in a logical calculus."

isabelle.in.tum.de

Isabelle [4] is a really powerful proof assistant coming with a higher order logic (HOL) proving environment. Isabelle proofs are written in the Isar ("intelligible semi-automated reasoning") language that is designed to make proofs readable and comprehensible for a mathematically inclined reader, with minimal overhead introduced by the formalism. In fact, "assistant" refers to the fact that the machine checks the proof provided by the user, in contrast to automatic theorem proving where the machine finds the proof itself. The tools for automation are intended to help the user write the proof at a conveniently high level, without needing to work at the level of a logical calculus, for example.

#### 2 Introduction

The objective of this project is to mechanize a proof of correctness of a set-based algorithm inspired by Tarjan's algorithm [5] for computing the strongly connected

components of a graph. A similar work has been done on Tarjan's algorithm [3]. The algorithm was first published in Vincent Bloemen's thesis [1] who furthermore gives and explains a few invariants, and was later reused in [2] with the aim of working on a parallel version of the algorithm.

In this report, a few arguments are given for the understanding of the formal proof. Some important invariants are explained and the main lemmas are briefly detailed. Some of the proofs will be explicitly given in slightly less rigorous mathematical terms for the sake of better understandability.

## 3 Formalisation

#### 3.1 Graphs and reachability

**Definition 1** (Directed graph). A directed graph  $\mathcal{G}$  is the data of a set of nodes  $\mathcal{V}$  and a set of oriented edges  $\mathcal{E}$ .

**Definition 2** (Reachability). For two vertices x and y of V, the reachability relation is noted " $\Rightarrow$ " such that  $x \Rightarrow^* y$  iff x can reach y in  $\mathcal{G}$ .

**Remark 1.** The relation  $\Rightarrow^*$  is in fact the transitive closure of the binary relation  $\Rightarrow$  defining edges in a graph.

**Definition 3** (Successors set for a node). Let  $\mathcal{G} = (\mathcal{V}, \mathcal{E})$  and  $v \in \mathcal{V}$ . The set of successors of  $v \in \mathcal{V}$  is Post(v) such that:

$$Post(v) = \{ w \in \mathcal{V} \mid (v, w) \in \mathcal{E} \}$$

## 3.2 Strongly connected components

#### 3.2.1 Directed graphs

**Definition 4** (SCC). Let  $\mathcal{G} = (\mathcal{V}, \mathcal{E})$  be a directed graph.  $\mathcal{C} \subseteq \mathcal{V}$  is a strongly connected component (SCC) of  $\mathcal{G}$  if:

$$\forall x, y \in \mathcal{C}, (x \Rightarrow^* y) \land (y \Rightarrow^* x)$$

i.e. there is a path between every x and y in C.

 $\mathcal{C}$  is maximal, or  $\mathcal{C}$  is a maximal SCC of  $\mathcal{G}$  if there is no other SCC containing  $\mathcal{C}$ , i.e. if:

$$\forall \mathcal{X}, (\mathcal{C} \subseteq \mathcal{X}) \land (\forall x, y \in \mathcal{X}, (x \Rightarrow^* y) \land (y \Rightarrow^* x)) \Longrightarrow \mathcal{C} = \mathcal{X}$$

**Definition 5.** (Strong connectedness) Let  $\mathcal{G} = (\mathcal{V}, \mathcal{E})$  be a directed graph.  $\mathcal{G}$  is strongly connected if  $\mathcal{V}$  is a SCC.

#### 3.2.2 Examples

Let us give some visual examples of a strongly connected component in a directed graph.



- (a) Strongly connected component
- (b) Not strongly connected component

Figure 1: Basic example of what is a small SCC

In Figure 1, two small directed graphs are shown. The first one (Figure 1a) is strongly connected, but the second one (Figure 1b) is not because the node x is not reachable from w for instance.




Figure 2: Example of a graph where each colored set of nodes is a – maximal – SCC

Let us give another example on a larger graph. FIGURE 2 shows a directed graph on which each colored set of nodes is a – maximal – SCC. Therefore, one can informally understand that SCCs roughly describe the cycles in a graph.




Figure 3: Reduced visualization of the graph represented in Figure 2

One may want to have a more general view of and consider only the distinct components of a graph. Thus, the graph show on Figure 2 can be reduced to a graph in which the previously colored nodes are replaced by a single node containing all the nodes of the same SCCs, *i.e.* all the equivalent nodes, as shown in Figure 3.

## 4 A sequential set-based algorithm

#### 4.1 Formalisation

**Definition 6** (SCC mapping). In the following algorithm, the SCCs are progressively tracked in a collection of disjoint sets through a map  $S: V \longrightarrow \mathcal{P}(V)$ , where  $\mathcal{P}(V)$  is the powerset of V, s.t. the following invariant is maintained:

$$\forall v, w \in \mathcal{V}, w \in \mathcal{S}(v) \iff \mathcal{S}(v) = \mathcal{S}(w) \tag{1}$$

**Remark 2.** In the following, the same notation S will be used to denote both the function defined above and the induced equivalence relation<sup>1</sup> since S associates to each node its class of equivalence.

<sup>&</sup>lt;sup>1</sup>For the relation  $(x, y) \mapsto x \in \mathcal{S}(y) \land y \in \mathcal{S}(x)$ 

Remark 3. In particular,  $\forall v \in \mathcal{V}, v \in \mathcal{S}(v)$ .

**Definition 7** (SCC union). Let UNITE be the function taking as parameters a map S as defined previously and two vertices u and v of V such that UNITE(S, u, v) merges the two mapped sets S(u) and S(v) and maintains the invariant (1) by updating S.

Let us give an example:

Let  $\mathcal{V} = \{u, v, w\}$  such that there is the following mapping:  $\mathcal{S}(u) = \{u\}$  and  $\mathcal{S}(v) = \mathcal{S}(w) = \{v, w\}$ .

Then, UNITE(S, u, v) =  $S(u) = S(v) = S(w) = \{u, v, w\}$ .

## 4.2 The algorithm

This section gives a pseudo-code of the set-based algorithm for which we will write a formal proof. See [1] for the original paper.

The algorithm only takes as input a directed graph  $\mathcal{G} = (\mathcal{V}, \mathcal{E})$  and a starting node  $v_0$  as the root of its exploration. It returns a partition SCCs of  $\mathcal{V}$  being the set of SCCs in the subgraph of  $\mathcal{G}$  reachable from  $v_0$ , where each element of SCCs is a maximal strongly connected components of  $\mathcal{G}$ . The equivalence relation  $\mathcal{S}$  is initialized so that at the beginning, each node is its own SCC. That is,  $\mathcal{S}(v) = \{v\}$  for all  $v \in \mathcal{V}$ , or seen as a set of disjoint sets:  $\mathcal{S} = \bigcup \{\{v\} \mid v \in \mathcal{V}\}$ .

#### Algorithm 1: Sequential set-based SCC algorithm

```
Data: \mathcal{G} = (\mathcal{V}, \mathcal{E}), v_0;
   Result: SCCs;
1 Initialize an empty set EXPLORED;
2 Initialize an empty set VISITED;
3 Initialize an empty stack R;
4 Initialize S: v \in \mathcal{V} \mapsto \{v\};
5 setBased(v_0);
6 function setBased: v \in V \rightarrow None
       VISITED := VISITED \cup \{v\};
7
       R.push(v);
8
       foreach w \in POST(v) do
9
           if w \in \textit{EXPLORED} then
10
                continue;
11
            end
12
            else if w \notin VISITED then
13
            | setBased(w);
14
            end
15
            else
16
                while S(v) \neq S(w) do
17
                    r := R.pop();
18
                    UNITE(S, r, R.top());
19
                end
20
           end
\mathbf{21}
       end
22
       if v = R. top() then
23
           report SCC S(v);
           \mathtt{EXPLORED} := \mathtt{EXPLORED} \cup \mathcal{S}(v);
25
           R.pop();
26
       end
\mathbf{27}
```

## 4.3 Informal proof

Note that this proof is said informal only because it is not checked by a mechanized proof assistant. Both logical and mathematical arguments developed below are absolutely relevant.

Lemma 1. (First invariant)

$$\forall x, y \in R, x \neq y \implies S(x) \cap S(y) = \varnothing$$

Note the misuse of the set notation  $x, y \in R$  which just means that x and y are in the stack R.

*Proof.* Let us consider the stacj R at some step of the algorithm. Let x = hd(R), i.e. the first element of the stack.

Then let  $y \in POST(x)$ . If y is already explored, then we can skip it.

Otherwise, if y is not in the visited set, then we explore it by calling the function setBased on y and since y is not visited,  $S(y) = \{y\}$ .

Finally, if y is in the visited set, it means that either y in on the stack or it has a representative  $z \in \mathcal{S}(y)$  on the stack. However, one can understand that the *while* loop below is executed until the equivalence relation is fully updated. In particular, the invariant is maintained, and there is only one representative of  $\mathcal{S}(y)$  on the stack, being the earliest visited node in  $\mathcal{S}(y)$  the traversal.

**Remark 4.** It is worth noting that the representative of a partial — with respect to the final result — SCC is the earliest node of this SCC to be visited regarding the graph traversal. This ensures that a representative on the stack is in fact the root of its SCC.

#### Lemma 2.

$$\biguplus_{r \in \mathbb{R}} \mathcal{S}(r) = \text{Live} := \text{Visited} \setminus \text{Explored}$$

*Proof.* The disjointness of all on-stack partial SCCs is given by lemma 1. Nodes from Visited \ Explored have a (unique) representative in R because they are being processed. So, Live  $\subseteq$  R.

By L.6-7 of Algorithm 1, VISITED  $\subseteq$  R. L.9-10 ensure that no explored node is pushed in R. L.24-25 keep the invariant by unstacking explored nodes from R, so  $R \cap Explored = \emptyset$ . Thus,  $R = VISITED \setminus Explored = Live$ .

Corollary 2.1 (Strong version).

$$\forall v \in \text{LIVE}, \exists ! \ r \in R \cap \mathcal{S}(v), \mathcal{S}(v) = \mathcal{S}(r)$$

Proof. Let  $v \in \text{LIVE} = \biguplus \mathcal{S}(r)$ . v is in a unique partial SCC  $\mathscr{S} := \mathcal{S}(v)$ . Because of lemma 1, there cannot exist  $x \neq y \in \mathbb{R}$  s.t.  $\mathcal{S}(x) = \mathcal{S}(y) = \mathscr{S}$ . Thus, there exists a unique  $x \in \mathbb{R}$  s.t.  $\mathcal{S}(x) = \mathscr{S}$  (and  $x \in \mathbb{R} \cap \mathscr{S}$ ).

Corollary 2.2 (Weak version).

$$\forall v \in \mathcal{V}, \forall w \in \text{Post}(v), w \in \text{Live} \implies \exists w' \in R, \mathcal{S}(w') = \mathcal{S}(w)$$

*Proof.* Holds because of corollary 2.1.

**Remark 5.** In the algorithm 1, this property is maintained by L.16-18. These lines also illustrate how the algorithm "reads" the SCCs. Corollary 2.2 shows that when the mapped representatives of the top two nodes of R are united (until S(w') = S(v) = S(w) since w' has a path to v), then all united components are in the same SCC.

Remark 6. Because R only contains exactly one representative for each partial SCC (corollary 2.1 and remark 4), after each step of the main loop – i.e. the DFS – every partial SCC is actually maximal in the current set of visited nodes.

**Theorem 1.** The sequential algorithm 1 is correct, i.e. it returns the set of maximal SCCs reachable from  $v_0$ .

```
Proof. Holds by remark 6.
```

An example of the execution of the algorithm is given in Appendix, in the section 6.2.

#### 4.4 Prerequisites for the formal proof

Since the informal proof seems to be convincing, the formal – checked automatically – proof can be written in Isabelle (HOL) based on the basis of the reasoning developed above.

#### 4.4.1 Environment setup

The first definitions should be the different structures used in the algorithm. In particular, a record containing all the sets needed and described in the pseudo-code of algorithm 1. The environment has a generic type parameter, which is used to represent the type of the nodes in the graph (often integers):

```
record 'v env =
    S :: "'v ⇒ 'v set"
    explored :: "'v set"
    visited :: "'v set"
    sccs :: "'v set set"
    stack :: "'v list"
```

The following lines define a graph structure and some useful natural relations:

```
locale graph = fixes vertices :: "'v set" and successors :: "'v \Rightarrow 'v set" assumes vfin: "finite vertices" and sclosed: "\forall x \in vertices. successors x \subseteq vertices"
```

The use of **successors** instead of an adjacency matrix, for instance, is a consequence of the fact that the algorithm is only concerned with the topological ordering of the nodes. For instance, nodes can represent integers, logical propositions or sets of states in a proving system for example.

#### 4.4.2 Reachability

Now that graphs are defined, the reachability can be defined. Defining an edge is simply some rewriting of being a successor of one node.

```
abbreviation edge where "edge x y \equiv y \in successors x"
```

Isabelle allows the definition of recursive functions. Such a definition must guarantee that any recursive call takes an argument that decreases (according to some measure function) with respect to the original argument. This is typically the case for functions that follow the definition of an algebraic type (integers, lists, trees, etc.). In addition, Isabelle allows inductive definitions of relations where two values are related if it is possible to justify this relation by applying a finite number of times the definition clauses of the relation. In our case, this applies to the definition of reachability where two nodes are connected if they are identical or if the first node has a successor from which the second node is already known to be reachable.

Although a recursive definition has to be based on some underlying inductive definition – which is simply not available for graphs – it expresses both the positive and negative information<sup>2</sup> whereas the inductive one only expresses the positive information directly. Therefore, with an inductive definition, the negative information has to be proved. One would be right to argue that it would be more convenient to be able to tell without proving it that two nodes are not reachable from each other, but this does not interest us for the following and this is actually more difficult than proving the reachability. Another important point is that there is no datatype for a recursive definition, especially in this case with the transitive closure of the  $\Rightarrow^*$  relation. Thus, the inductive definition is not a choice but is necessary. Isabelle provides a construction for inductive predicate definitions, which is appropriate here because the two clauses represent the reflexive case and the extension of reachability by prepending an edge. This will be particularly useful in the proofs.

<sup>&</sup>lt;sup>2</sup>In this case, the positive information designates the fact of being reachable and the negative information designates the fact of not being reachable.

```
inductive reachable where
  reachable_refl[iff]: "reachable x x"
| reachable_succ[elim]: "[edge x y; reachable y z] \impress reachable x z"
```

In order to be able to use those relations in the proofs later, it is essential to prove a list of lemmas, namely all the different natural properties that Isabelle cannot deduce<sup>3</sup> from nothing<sup>4</sup>. For instance, the following lemmas are essential.

```
lemma succ_reachable:
```

```
assumes "reachable x y" and "edge y z" shows"reachable x z" using assms by induct auto
```

Mathematical writing:  $\forall x, \forall y, \forall z, (x \Rightarrow^* y \land y \Rightarrow z) \Longrightarrow x \Rightarrow^* z$ 

Remark 7. Note that this is the "mirror" of clause reachable\_succ (appending an edge).

```
lemma reachable_trans:
```

```
assumes y: "reachable x y" and z: "reachable y z" shows "reachable x z" using assms by induct auto
```

Mathematical writing:  $\forall x, \forall y, \forall z, (x \Rightarrow^* y \land y \Rightarrow^* z) \Longrightarrow x \Rightarrow^* z$ 

As the formal proofs will enventually deal with strongly connected components, it is also essential to formally define SCCs. For the purpose of the proof, the property of being a SCC is called sub\_scc and being a maximal SCC is called is\_scc:

```
definition is_subscc where
```

```
"is_subscc S \equiv \forall x \in S. \forall y \in S. reachable x y"
```

Mathematical writing: A set S is a SCC if  $\forall x \in S, \forall y \in S, x \Rightarrow^* y$ 

```
{\tt definition} \ {\tt is\_scc} \ {\tt where}
```

```
"is_scc S \equiv S \neq {} \land is_subscc S \land (\forall S'. S \subseteq S' \land is_subscc S' \longrightarrow S' = S)"
```

Mathematical writing: A non-empty SCC S is maximal if for all SCC S',  $S \subseteq S' \Longrightarrow S' = S$ 

<sup>&</sup>lt;sup>3</sup>That is an abuse of language. The idea is for example that for the moment, there is no formal link between edge and reachable. The goal is to formalize it so Isabelle is logically able to both use and simplify some results in the proofs.

<sup>&</sup>lt;sup>4</sup>There is actually a theorem fetcher that is particularly useful to find a basic set of lemmas.

Once again, there are some lemmas to prove which are deduced using Isabelle from the above definitions, such as giving conditions on when an element can be added to a SCC, or that two vertices that are reachable from each other are in the same SCC, or that two SCCs having a common element are identical, etc.

#### 4.4.3 Ordering relation

In the proof, a precedence relation<sup>5</sup>noted •  $\leq$  • in • will be needed on the stack. Let x and y be two nodes and R be a stack. Informally, x precedes y in R if y was pushed in R before x (see Figure 4).




Figure 4: The ordering relation on stacks

**Definition 8** (Ordering relation). Let x and y be two nodes and xs be a stack.

$$x \leq y \text{ in } xs \equiv \exists h, \exists r, (xs = h@[x]@r) \land (y \in [x]@r)$$

The idea is to later use the following property: if  $x \leq y$  in xs, then  $y \Rightarrow^* x$ . It is defined in Isabelle as follows:

```
definition precedes ("\_ \le \_ in \_" [100,100,100] 39) where "x \le y in xs \equiv \existsh r. xs = h @ (x # r) \land y \in set (x # r)"
```

All the different properties (i.e. lemmas) which follow this definition in the Isabelle implementation are detailed in the natural mathematical writing in the Appendix. The right part of the notation represents the orders of priority for each operand since  $\leq$  is an infix operator.

#### 4.4.4 Implementation of the algorithm

In the algorithm 1, the SCCs are progressively tracked in sccs and the equivalence relation S is updated with the UNITE function. Note that the function written in Isabelle is different from the UNITE function introduced earlier. From now on,

<sup>&</sup>lt;sup>5</sup>In fact, a partial order is being defined on stacks.

unite designates the following function, which was first written as a recursive function as follows:

```
function unite :: "'v \Rightarrow 'v env \Rightarrow 'v env" where

"unite v w e =

(if (S e v = S e w) then e

else let r = hd(stack e);

r'= hd(tl(stack e));

joined = S e r \cup S e r;

e'= e(|

stack := tl(stack e),

S := (\lambda n. if n \in joined then joined else S e n)

in unite v w e')"

by pat_completeness auto
```

However, this definition makes the proofs too difficult due to the recursion. A non-recursive version was therefore written:

```
definition unite :: "'v \Rightarrow 'v \Rightarrow 'v env \Rightarrow 'v env" where
"unite v w e \equiv
let pfx = takeWhile (\lambdax. w \notin \mathcal{S} e x) (stack e);
sfx = dropWhile (\lambdax. w \notin \mathcal{S} e x) (stack e);
cc = \bigcup \{\mathcal{S} \in x \mid x. \quad x \in \text{set pfx} \cup \{\text{hd sfx}\}\}
in e(\mathcal{S} := \lambdax. if x \in \text{cc then cc else } \mathcal{S} \in x, stack := sfx)"
```

The idea of this definition is to create a partition of stack e = pfx @ sfx such that pfx contains the nodes which are to be merged into S = w and sfx contains the root of S = w followed by the rest of the stack. Then, cc - w which stands for connected component – contains all the nodes which are equivalent to w in the sub-graph currently explored. The function takeWhile applied to a boolean function  $P^6$  seen as a property and a list xs returns the elements of xs which satisfy P and stops at the first element not satisfying P. The function dropWhile is the opposite of takeWhile. Both the recursive and non-recursive versions of unite are equivalent – it should be proved though – in the context of this report. In fact, the non-recursive definition is intended to be simpler for the proof because it avoids introducing separate pre- and post-conditions for the function and proving such a "contract".

Now that the environment is set up, the actual algorithm – seen as a function – can be implemented. Since Isabelle does not support loops, the implementation will be split into two mutually recursive functions. The main function is called dfs

 $<sup>^6</sup>$ P :: 'a  $\Rightarrow$  bool

and takes its name after the Depth First Search algorithm because the algorithm 1 roughly consists in a deep traversal of a graph. The second function is called dfss and represents the *foreach* loop of the algorithm 1. The two functions are mutually recursive because they recursively call each other. In particular, dfss will call both dfs and itself, depending on the case. Their implementation is as follows:

```
"'v \Rightarrow v env \Rightarrow v env" and
function dfs ::
                  "'v \Rightarrow v set v = v env v = v env" where
         dfss::
"dfs v e =
    (let e1 = e(visited := visited e \cup {v}, stack := (v # stack e));
         e' = dfss v (successors v) e1
    in if v = hd(stack e')
         then e'(|sccs:=sccs e' \cup S e' v, explored:=explored e' \cup (S e' v),
stack:=tl(stack e')
         else e')"
| "dfss v vs e =
    (if vs = \{\} then e
    else (let w = SOME x. x \in vs
         in (let e' = (if w ∈ explored e then e
                   else if w \notin visited e then dfs w e
                   else unite v w e)
              in dfss v (v - \{w\}) e')))"
  by pat_completeness (force+)
```

The two last keywords require explanations as well: pat\_completeness stands for pattern completeness and ensures that there is no missing patterns. The method force finishes the proof of pattern completeness, the proof of termination remains open, and it would actually show that these are well-defined functions. force is more aggressive in instantiation than auto and seems to find the right instance.

#### 4.4.5 General scheme

As the algorithm is composed of two mutually recursive functions, the correctness of the algorithm is proved by mutual induction on the functions with the help of the environment structure (cf 4.4.1). Since both dfs and dfss are quite complex, the proof is split into several parts. The idea is to prove for each function that its execution given some pre-conditions on the input environment implies some post-conditions on the output environment. Then, it has to be made for the mutually recursive calls as well, so that given the same pre-conditions on one function, the pre-conditions on the other function are also satisfied. Finally, it has to be proved that if the pre-conditions are satisfied for one function, and if the pre-conditions

imply the post-conditions on the other function, then the post-conditions are also satisfied for the first function.

#### 4.4.6 Well-formedness of the environment

The whole proof relies on one big invariant regarding the environment structure. It defines the fact for an environment to be well-formed. This invariant is a conjunction of several properties and is defined as follows:

```
definition wf_env where

"wf_env e \equiv
    distinct (stack e)
    \land set (stack e) \subseteq visited e
    \land explored e \subseteq visited e
    \land explored e \cap set (stack e) = {}
    \land (\forall v w. w \in S e v \longleftrightarrow (S e v = S e w))
    \land (\forall v \in set (stack e).\forall w \in set (stack e).v \neq w \longleftrightarrow S e v \cap S e w = {})
    \land (\forall v. v \notin visited e \longleftrightarrow S e v = {v})
    \land (\forall v. v \notin visited e \longleftrightarrow S e v = {v})
    \land (\forall x y. x \preceq y in stack e \longleftrightarrow reachable y x)
    \land (\forall x. is_subscc (S e x))
    \land (\forall x \in explored e. \forall y. reachable x y \longleftrightarrow y \in explored e)
    \land (\forall S \in sccs e. is_scc S)"
```

Let us take a closer look to this invariant, taken in the same order as the definition above:

- First, the stack is a list of distinct elements.
- All elements of the stack are visited.
- The set of explored nodes is a subset of the set of visited nodes.
- Explored nodes cannot be in the stack.
- The three next properties are about the equivalence relation  $\mathcal{S}$ .
- The union of the sets of equivalent nodes in the stack is equal to the set of visited nodes minus the set of explored nodes.
- A node in the stack can reach all nodes before it in the stack (*i.e.* pushed later).
- $\bullet$  S represents a set of strongly connected components (not maximal).

- For all explored nodes, the sub-graph induced by their successors is totally explored.
- sccs is a set of maximal SCCs

These properties are natural and most of them are easy to prove. Actually, there is a bit of redundancy here. For example, the second and fourth conjunct follow from the fifth and eighth. This is not a bad thing per se since it may help automatic proof, but could be discussed.

It is also useful to induce a notion of monotonicity on the environments during the execution of the algorithm. This is defined as follows through the definition of an ordering relation on environments:

In particular, the last conjunct expresses the fact that the equivalence relation on the stack is monotonic. This is a useful property as it explicitly gives a monotonic behaviour on the environment structure – hence its name – during the execution of the algorithm, even though we do not have an inductive rule on environments. Besides, this property is transitive, which will be used in the proofs.

#### 4.4.7 dfs pre- and post-conditions

The pre-conditions of dfs are rather simple. The environment must be well-formed and the node must not be visited. There is also a condition on the reachability of the nodes in the stack and the node on which the function is called, but once again this condition is rather natural to consider since dfs is performing a DFS graph traversal:

```
definition pre_dfs where
  "pre_dfs v e ≡
     wf_env e
     ∧ v ∉ visited e
     ∧ (∀ n ∈ set (stack e). reachable n v)"
```

The post-conditions are a little more complex since it has to consider the new environment with the new visited / explored nodes, the new state of the stack and the updates in S:

```
definition post_dfs where

"post_dfs v prev_e e \equiv
    wf_env e

\land (\forall x. \text{ reachable } v x \longrightarrow x \in \text{ visited } e)

\land \text{ sub\_env prev\_e } e

\land (\forall n \in \text{ set } (\text{stack } e). \text{ reachable } n v)

\land (\exists \text{ ns. stack prev\_e} = \text{ ns } @ (\text{stack } e))

\land (\forall \text{ m n. m } \preceq \text{ n in prec\_e } \longrightarrow

(\forall \text{ u } \in \mathcal{S} \text{ prev\_e m. reachable } \text{ u } v \land \text{ reachable } v \text{ n } \longrightarrow \mathcal{S} \text{ e m } = \mathcal{S} \text{ e n}))

\land ((\text{v } \in \text{ explored } e \land \text{ stack } e = \text{ stack prev\_e}) \lor

(\text{v } \in \mathcal{S} \text{ e } (\text{hd } (\text{stack } e))) \land

(\exists \text{ n } \in \text{ set } (\text{stack prev\_e}). \mathcal{S} \text{ e } v = \mathcal{S} \text{ e n}))"
```

Let us give some explanations:

- The environment must be well-formed.
- The sub-graph induced by the nodes reachable from the node on which the function is called must be totally visited.
- The previous environment must be a sub-environment of the new one (*i.e.* there is a monotonic ordering on the environments).
- The condition of reachability on the stack from the pre-condition must remain satisfied.
- The new stack is a suffix of the previous one (*i.e.* it expressively represents the structure of FIFO stack).
- The last one is easier to understand: either v is explored and the stack considered before and after the execution of the function has not changed, or the head of the stack is the representative of v and v had a representative in the previous stack.

#### 4.4.8 dfss pre- and post-conditions

The pre-conditions on dfss are also rather simple, except the sixth conjunct which expresses the fact that before the execution of the function, if a node n on the stack is reachable from v, then either  $v \in \mathcal{S}(n)$ , or n is reachable from a successor of v. Once again, this condition has to be equated with the fact that all nodes on stack can reach v (fifth conjunct in the following definition). The definition is as follows:

The post-conditions on dfss are less complicated than the post-conditions of dfs and can easily be understood with the explanations of the other invariants:

```
definition post_dfss where "post_dfss v vs prev_e e \equiv wf_env e \land (\forall w \in vs. \forall x. reachable w x \longrightarrow x \in visited e) \land sub_env prev_e e \land (\forall n \in set (stack e). reachable n v) \land (stack e \neq []) \land (\forall n \in set (stack e). reachable v n \longrightarrow v \in S e n) \land (\exists ns. stack prev_e = ns @ (stack e)) \land (v \in S e (hd (stack e)))"
```

#### 4.5 Formal proof

#### 4.5.1 pre\_dfs implies pre\_dfss

This lemma assumes that the pre-conditions on  $\mathtt{dfs}$  are satisfied and shows that it implies that the pre-conditions on  $\mathtt{dfss}$  are satisfied on the environment on which  $\mathtt{dfs}$  is called, *i.e.* the environment for which v was pushed on the stack and added to the set of visited nodes.

```
lemma pre_dfs_pre_dfss:
   assumes "pre_dfs v e"
   shows "pre_dfss v (successors v)
        e(visited := visited e U {v}, stack := v # stack e)"
```

The proof is quite straightforward, except for the reachability of elements in the stack. Let e' be the environment on which dfss will be called. Then, we want to show the following proposition:

$$\forall x, y, \ x \leq y \text{ in stack } e' \Longrightarrow y \Longrightarrow^* x$$

*Proof.* Let x and y be two nodes on the stack of e' such that  $x \leq y$  in stack e'.

if x = v: then we have to show that  $y \Rightarrow^* v$ . If y = v, it is trivial, otherwise, y is in the stack of e and v is reachable from y from the pre-condition of dfs (fourth conjunct in the definition of dfs).

if  $x \neq v$ : then x is in the stack of e and  $x \leq y$  in stack e. From the pre-condition of dfs,  $y \Rightarrow^* x$ .

#### 4.5.2 pre\_dfss implies pre\_dfs

This lemma fixes a node w and assumes the pre-conditions on dfss on a node v, a set (of successors) vs and an environment e. It assumes that w is a successor of v that is not visited, and it shows that the pre-conditions on dfs on w the same environment e are satisfied.

```
lemma pre_dfss_pre_dfs:
    fixes w
    assumes "pre_dfss v vs e" and "w \notin visited e" and "w \in vs"
    shows "pre_dfs w e"
```

The proof of the lemma is very simple since all conditions of pre\_dfs can be deduced from the conditions of pre\_dfss.

#### 4.5.3 pre\_dfs implies post\_dfs

This lemma is probably the most important lemma – and the most difficult to show – in the proof of the algorithm. It assumes that the pre-condition on dfs are satisfied and shows that the post-condition on dfs are satisfied. It is stated as follows:

lemma pre\_dfs\_implies\_post\_dfs:

```
fixes v e defines "e1 \equiv e(visited := visited e \cup {v}, stack := (v # stack e))" defines "e' \equiv dfss v (successors v) e1" assumes 1: "pre_dfs v e" and 2: "dfs_dfss_dom (Inl(v, e))" and 3: "post_dfss v (successors v) e1 e'" shows "post_dfs v e (dfs v e)"
```

Note that the definitions of  $e_1$  and e' follow from the definition of dfs which relies – in an if statement – on the fact that v = hd (stack e) or not, therefore the proof also considers this split of cases.

The first case assumes that v = hd(stack e'). Then with the second assumption (v and e are in the domain of definition of dfs), we can show that dfs returns the following environment:

```
dfs v \ e = e'(|\operatorname{sccs} := \operatorname{sccs} e' \cup \{\mathcal{S} \ e' \ v\}, 

\operatorname{explored} := \operatorname{explored} \ e' \cup (\mathcal{S} \ e' \ v), 

\operatorname{stack} := \operatorname{tl}(\operatorname{stack} \ e')|)"
```

That is, it returns an environment  $e_2 := \text{dfs } v \text{ } e \text{ obtained from } e' \text{ in which a new SCC, namely } \mathcal{S} \text{ } e' \text{ } v, \text{ is added.}$  The different conjuncts in the definition of post\_dfs are then shown one by one. Some useful things to note – and prove – are the following:

• stack  $e_2 = \text{tl}(\text{stack } e')$ , and since v is the head of the stack of e', we can show that stack  $e_1 = \text{stack } e'$ . The main idea is to use, from the assumption 3, the fact that:

$$\exists ns, \text{ stack } e_1 = ns @ (\text{stack } e')$$

- The equivalence relation is the same for e' and  $e_2$ :  $\forall x, \mathcal{S} \ e' \ x = \mathcal{S} \ e_2 \ x$
- It is easy to show that e is a sub-environment of  $e_1$  (i.e.  $sub\_env \ e \ e_1$ ). We can also show that  $e_1$  is a sub-environment of  $e_2$ , which is much more difficult. However, the proofs in Isabelle are quite simple to read and understand. In fine, by transitivity of the relation  $sub\_env$ , e is a sub-environment of  $e_2$ .
- Showing that S e' v is a maximal SCC in the environment  $e_2$  is a tricky part. The proof uses a contradiction-based reasoning and the fact that the equivalence relation is the same for e' and  $e_2$  to find a node being in a bigger SCC containing S e' v but not in S e' v, and exploit the reachability from and to v and the assumption 3.

In the second case, v is not the head of the stack of e', therefore  $dfs\ v\ e=e'$  and the proof is much easier thanks to the post-condition on dfss, *i.e.* assumption 3.

#### 4.5.4 Partial correctness

This lemma shows two things:

```
dfs_dfss_dom (Inl(v,e)) \land pre_dfs v e \Longrightarrow post_dfs v e (dfs v e) dfs_dfss_dom (Inr(v,e)) \land pre_dfss v vs e \Longrightarrow post_dfss v vs e (dfss v vs e) where dfs_dfss_dom (Inl(v,e)) - respectively Inr(v, e) - is the assumption that v and e are in the domain of definition of dfs (respectively dfss).
```

The first proposition is a consequence of the lemma pre\_dfs implies post\_dfs.

For the second proposition, we need to write the proof according to the definition of dfss. That is, we need to show the following proposition:

With these assumptions:

- $dfs_dfs_dom(Inr(v, vs, e))$
- ullet pre\_dfss  $v\ vs\ e$

 $\bullet \ vs \neq \varnothing \ \text{and} \\ \forall w \in vs, w \notin \texttt{explored} \ e \wedge w \notin \texttt{visited} \ e \wedge \texttt{pre\_dfs} \ w \ e \Longrightarrow \texttt{post\_dfs} \ w \ e \ (\texttt{dfs} \ w \ e)$ 

• let 
$$e' = \begin{cases} e & w \in \text{explored } e \land w \notin \text{visited } e \land \text{predits } w \in \text{otherwise} \\ \text{dfs } v \ e & w \in \text{visited } e \\ \text{unite } v \ w \ e & \text{otherwise} \end{cases}$$

$$vs \neq \varnothing \text{ and}$$

$$\forall w \in vs, \text{pre_dfss } v \ (vs - \{w\}) \ e' \implies \text{post_dfss } v \ (vs - \{w\}) \ e' \text{ (dfss } v \ (vs - \{w\}) \ e')$$

We need to show the post-condition of dfss:

$$post\_dfss \ v \ vs \ e \ (dfss \ v \ vs \ e)s$$

If the set of successors vs is empty, then the post-condition of dfss is satisfied. Otherwise, the proof is split according to the cases in the definition of e'. In the first case, w is in the set of explored nodes of e. Hence, e' = e. The proof is rather straightforward thanks to the assumption on pre\_dfss.

The following case, i.e.  $w \in \text{visited } e$  is much more complicated. In this case, e' = dfs v e. Although the proof remains straightforward, the different properties are more difficult to establish. The main idea is to use the fact that post\_dfs w e e' is satisfied to show the pre-condition: pre\_dfss v  $vs - \{w\}$  e. From the assumptions, we can obtain the post-condition: post\_dfss v  $vs - \{w\}$  e e'. There still remains some work to do inorder to show that this implies the final post-condition, i.e. that post\_dfss v  $vs - \{w\}$  e (dfss v  $vs - \{w\}$  e) is satisfied.

In the last case, the function unite is called to compute the new environment e'. The proof uses the local variables defined in the definition of unite, *i.e.* pfx, sfx and cc (see 4.4.4). In particular, it is important to note that stack e = pfx @ sfx and w is represented by a node in the stack being the head of sfx. A visual representation of the stack is given in Figure 5. Morally, it is easy to understand that because of the cycle drawn on Figure 5, S should be updated in the new environment so that all components of S for each node of sfx should be merged together.

The approach is similar to the previous case: we prove the pre-condition on dfss and use the hypotheses to show the post-condition of dfss. The rest consists in proving the post-condition of dfss on v,  $vs - \{w\}$  and the new environment e' and the environment return by dfss implies the final post-condition.

#### 5 Conclusion

### 5.1 What has been done?

The proof of correctness is almost done. The model and the invariants we have come up with are quite comprehensive and coherent now. We regularly test newer invariants with TLC, a model checker for specifications written in TLA+. Thus, we also have a TLA+




Figure 5: Representation of stack e in the case  $w \in \text{visited } e$ . The black arrows represent reachability in both directions because the definition of a representative on the stack states that both nodes are in the same component of S for the environment e.

version of this algorithm, although it is not really the same implementation. However, as we juste use it to verify a few properties on specific examples, it is just meant to give us a better feeling of confidence or find counter examples. I've been working on this project for almost a year, and yet it's only been a month or two that I really feel like I understand the algorithm.

## 5.2 What is missing?

In the proof of correctness, some intermediate proofs are still skipped. It is mainly due to some former invariants that were deleted because they were wrong.

Another important missing piece is the proof of termination of the functions. In order to achieve this, we need to define a measure and find a quantity that strictly decreases for that measure. In sub\_env, the inclusions are not strict. An idea would be to consider the complementary set of visited between two calls of dfs as a set getting strictly smaller.

## 5.3 My experience

Before starting the project, I had no knowledge of formal methods. I had to learn it quickly in one month and as I went along. I liked it right away. As I am very attached to rigour, there is a very addictive side to wanting to demonstrate everything. This is why I decided to pursue my career in the field of formal methods.

It was also my first experience in the world of research. It was an opportunity to confirm my desire to do basic research, and I also found a domain. I was also able to meet several of the researchers of the MOSEL-VERIDIS team at LORIA in Nancy, as well as Jaco Van de Pol who is the supervised the thesis of Vincent Bloemen [1].

## 6 Appendix

#### 6.1 Some lemmas

Those lemmas refer to the precedence relation introduced in Section 4.4.3.

Let x, y, z be three nodes, and let xs, ys, zs be three lists of nodes representing stacks. By abuse of language, if an element is on a stack, it is in the set of elements contained in the stack so the following statement can be written: x is on  $xs \iff x \in xs$ . However, xs in not seen as the set representing xs since an element may occur several times in a stack. The operator @ denotes the concatenation and operates on two lists:  $[x_0, \ldots, x_n]@[y_0, \ldots, y_m] = [x_0, \ldots, x_n, y_0, \ldots, y_m]$ .

(i) 
$$x \leq y$$
 in  $xs \Longrightarrow (x \in xs) \land (y \in xs)$ 

(ii) 
$$y \in [x]@xs \Longrightarrow x \leq y$$
 in  $([x]@xs)$ 

(iii) 
$$x \neq z \Longrightarrow (x \leq y \text{ in } ([z]@zs) \Longrightarrow x \leq y \text{ in } zs)$$

(iv) 
$$(y \leq x \text{ in } (\lceil x \rceil @xs)) \land (x \notin xs) \Longrightarrow (x = y)$$

(v) 
$$y \in (ys@[x]) \Longrightarrow y \leq x$$
 in  $(ys@[x]@xs)$ 

(vi) 
$$(x \leq x \text{ in } xs) = (x \in xs)$$

(vii) 
$$x \leq y$$
 in  $xs \Longrightarrow x \leq y$  in  $(ys@xs)$ 

(viii) 
$$x \notin ys \Longrightarrow (x \leq y \text{ in } (ys@xs) \Longleftrightarrow x \leq y \text{ in } xs)$$

(ix) 
$$x \prec y$$
 in  $xs \Longrightarrow x \prec y$  in  $(xs@ys)$ 


(x) 
$$y \notin ys \Longrightarrow x \leq y$$
 in  $(xs@ys) \Longleftrightarrow x \leq y$  in  $xs$ 

$$(x \leq y \text{ in } xs) \land (y \leq z \text{ in } xs) \land \underbrace{(\forall \ 0 \leq i < j \leq \text{length}(xs), xs[i] \neq xs[j])}_{\text{all elements of } xs \text{ are distinct}} \Longrightarrow x \leq z \text{ in } xs$$


$$(x \leq y \text{ in } xs) \land (y \leq x \text{ in } xs) \land \underbrace{(\forall \ 0 \leq i < j \leq \text{length}(xs), xs[i] \neq xs[j])}_{\text{all elements of } xs \text{ are distinct}} \Longrightarrow x = y$$

## 6.2 An example of execution of algorithm 1.


This sequence of figures is meant to be read from left to right. Brown nodes are nodes which have not been visited yet. Pink nodes are nodes which are being visited, *i.e.* which are on stack. Other colors are for reported SCCs.



 $\mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\} \\ \mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\} \\ \mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{B\} \cup \{C\} \cup \{$ 




R = [H, E] $VISITED = \{H, E\}$  $DEAD = \{\}$ 




R = [H, E, A, B] $VISITED = \{H, E, A, B\}$  $DEAD = \{\}$ 

 $\mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\}$ 




R = [H] $VISITED = \{H\}$  $DEAD = \{\}$ 




R = [H, E, A] $VISITED = \{H, E, A\}$  $DEAD = \{\}$ 

 $\mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\} \\ \mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\} \\ \mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{B\} \cup \{C\} \cup \{$ 




R = [H, E, A, B, C] $VISITED = \{H, E, A, B, C\}$  $DEAD = \{\}$ 

 $\mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\}$ 




R = [H, E, A, B, C, D]  $VISITED = \{H, E, A, B, C, D\}$   $DEAD = \{\}$ 

 $\mathcal{S} = \{A\} \cup \{B\} \cup \{C\} \cup \{D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\}$ 




R = [H, E, F]  $VISITED = \{H, E, A, B, C, D, F\}$   $DEAD = \{A, B, C, D\}$ 

 $\mathcal{S} = \{A,B,C,D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\}$ 




R = [H, E, F]  $VISITED = \{H, E, A, B, C, D, F, G\}$ 

 $\begin{aligned} \text{DEAD} &= \left\{ \text{A, B, C, D, G} \right\} \\ \mathcal{S} &= \left\{ \text{A, B, C, D} \right\} \cup \left\{ \text{E} \right\} \cup \left\{ \text{F} \right\} \cup \left\{ \text{G} \right\} \cup \left\{ \text{H} \right\} \end{aligned}$ 




R = [H, E]  $VISITED = \{H, E, A, B, C, D\}$   $DEAD = \{A, B, C, D\}$ 

 $\mathcal{S} = \{A,B,C,D\} \cup \{\underbrace{E}\} \cup \{F\} \cup \{G\} \cup \{H\}$ 



 $R = [H, \, E, \, F, \, G]$   $VISITED = \{H, \, E, \, A, \, B, \, C, \, D, \, F, \, G\}$   $DEAD = \{A, \, B, \, C, \, D\}$ 

 $\mathcal{S} = \{A,B,C,D\} \cup \{E\} \cup \{F\} \cup \{G\} \cup \{H\}$ 



R = []  $VISITED = \{H, E, A, B, C, D, F, G\}$ 

 $\begin{aligned} \text{DEAD} &= \left\{ \text{A, B, C, D, G, E, F, H} \right\} \\ \mathcal{S} &= \left\{ \text{A, B, C, D} \right\} \cup \left\{ \text{E, F, H} \right\} \cup \left\{ \text{G} \right\} \end{aligned}$ 

## References

- [1] V. Bloemen. Strong connectivity and shortest paths for checking models. PhD, University of Twente, Enschede, The Netherlands, July 2019. ISBN: 9789036547864.
- [2] Vincent Bloemen, Alfons Laarman, and Jaco van de Pol. Multi-core on-the-fly SCC decomposition. *ACM SIGPLAN Notices*, 51(8):8:1–8:12, February 2016.
- [3] Ran Chen, Cyril Cohen, Jean-Jacques Lévy, Stephan Merz, and Laurent Théry. Formal Proofs of Tarjan's Strongly Connected Components Algorithm in Why3, Coq and Isabelle. page 18.
- [4] Tobias Nipkow, Lawrence C. Paulson, and Markus Wenzel. *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*, volume 2283 of *Lecture Notes in Computer Science*. Springer Berlin / Heidelberg, Berlin, Heidelberg, 2002. Book Title: Isabelle/HOL.
- [5] Robert Endre Tarjan. Depth-first search and linear graph algorithms. SIAM J. Comput., 1(2):146–160, 1972.